Wolfram Technology Conference 2010

Modeling and Simulation of French Elections

Yves A. Papegay / INRIA Sophia Antipolis Méditerranée

Abstract

First round of the french regional elections took place on March 14th, 2010, and second round on March 21st. The official results were available online very few time after the end of the consultation. During the week between the two rounds, surveys were very popular, as well as public debates, analyses of strategies of alliances, and people were very excited in showing statistics, trying to predict or discover what will be the final results.

Instead of using more or less well founded statistical approaches to forecast the results of the second round, we developped a complete model at the granularity of municipalities based on similar elections in 2004. Our guess was that at this level the reactions of people to political alliances should be very similar from one election to the next one.

We will described our approaches and implementations, and demonstrated that *Mathematica* is really a valuable tool for performing very easily the necessary tasks of such a modeling and simulation process : acquiring data, checking it, analyzing it, building models, prototyping simulation engines, and visualizing the results.

In the current talk, we are focusing on the models themselves. We will describe namely:

- how a textual analysis of the lists of candidates could provide heuristics to derive the models,
- how models can help in verifying political hypothesis on what has an influence on votations,
- how uncertainty in models can be held with the help of intervals
- how the developed framework could be easily extended to analyse and/or predict results of elections everywhere in the world

Presentation

Motivations

- an original application of modeling and simulation methods
- case study of decent size with available data
- big potential of application

Regional Elections

- Two-rounded elections held every 6 years
- 43 millions of people electing 1880 counceilors in 26 regions
- List votation with merging between the 2 rounds

Results

- Published on the web
- Given in terms of regions (26), departments (100) and communes (36718)

- Given in terms of votes and of number of elected people, for list of candidates and political families

Data Acquisition

Web - Based : “SLAMing” the site (75)

Structure

Pages

Global

Local



Data Model

Datas (100)



Results (75)

France | Regions | Departments



Seats



Data Curating

A vote can not be lost

At France level

```
CheckResult[year_, round_, "FRANCE"] :=
 FranceResult[year, round, "Inscrits"] - (FranceResult[year, round, "Abstentions"] + FranceResult[year,
       round, "Blancs ou nuls"] + Total[Map[FranceResult[year, round, #] &, familyCodes[year, round]]]) == 0

Outer[CheckResult[#1, #2, "FRANCE"] &, {"04", "10"}, {"R1", "R2"}]
Position[%, Except[True], {3}, Heads → False]
```

```
{{True, True}, {True, True}}
```

```
{}
```

At Region level

```
CheckRegionResult[year_, round_, reg_] :=
 If[ListQ[Candidat[year, round, reg]], RegionResult[year, round, "Inscrits", reg] -
     (RegionResult[year, round, "Abstentions", reg] + RegionResult[year, round, "Blancs ou nuls", reg] +
       Total[Map[RegionResult[year, round, #, reg] &, Candidat[year, round, reg]]]) == 0, True]
```

```
Outer[CheckRegionResult, {"04", "10"}, {"R1", "R2"}, regionCodes[]];
Position[%, Except[True], {3}, Heads → False]
```

```
{}
```

At Department level

```
CheckDepartmentResult[year_, round_, dept_] :=
 If[ListQ[Candidat[year, round, Region[dept]]], DepartmentResult[year, round, "Inscrits", dept] -
    (DepartmentResult[year, round, "Abstentions", dept] + DepartmentResult[year, round, "Blancs ou nuls", dept] +
      Total[Map[DepartmentResult[year, round, #, dept] &, Candidat[year, round, Region[dept]]]]) == 0, True]
```

```
Outer[CheckDepartmentResult, {"04", "10"}, {"R1", "R2"}, departmentCodes[]];
Position[%, Except[True], {3}, Heads → False]
```

```
{}
```

At Commune level

```
CheckCommuneResult[year_, round_] := Map[CheckCommuneResult[year, round, #] &, communeCodes[year]]
```

```
CheckCommuneResult[year_, round_, comm_] :=
 If[ListQ[Candidat[year, round, Region[Department[comm]]]], CommuneResult[year, round, "Inscrits", comm] -
    (CommuneResult[year, round, "Abstentions", comm] + CommuneResult[year, round, "Blancs ou nuls", comm] +
      Total[Map[CommuneResult[year, round, #, comm] &, Candidat[year, round, Region[StringTake[comm, 3]]]]]) == 0, True]
```

```
Outer[CheckCommuneResult, {"04", "10"}, {"R1", "R2"}];
Position[%, Except[True], {3}, Heads → False]
```

```
{}
```

Geographical Consistency

Between France and Regions

```
Clear[check]
check[year_, round_] := Map[check[year, round, #] &,
  Join[{"Inscrits", "Abstentions", "Blancs ou nuls", "Exprimés", "Votants"}, familyCodes[year, round]]]
check[year_, "R1", res_] := FranceResult[year, "R1", res] ==
  Total[Map[RegionResult[year, "R1", res, #] &, regionCodes[year]]]
check[year_, "R2", res_] := FranceResult[year, "R2", res] ==
  Total[Map[RegionResult[year, "R2", res, #] &, Select[regionCodes[year], ListQ[Candidat[year, "R2", #]] &]]]
```

```
Outer[check[#1, #2] &, {"04", "10"}, {"R1", "R2"}];
Position[%, Except[True], {3}, Heads → False]
```

```
{}
```

Between Regions and Departments

```
Clear[check]
check[year_, round_] := Map[check[year, round, #] &, regionCodes[year]]
check[year_, round_, region_] := If[ListQ[Candidat[year, round, region]], Map[check[year, round, #, region] &,
   Join[{"Inscrits", "Abstentions", "Blancs ou nuls", "Exprimés", "Votants"}, Candidat[year, round, region]]], {True}]
check[year_, round_, res_, region_] := RegionResult[year, round, res, region] ==
  Total[Map[DepartmentResult[year, round, res, #] &, Department[region]]]
```

```
Outer[check[#1, #2] &, {"04", "10"}, {"R1", "R2"}];
Position[%, Except[True], {4}, Heads → False]
```

```
{}
```

Between Departments and Communes

```
Clear[check]
check[year_, round_] := Map[check[year, round, #] &, departmentCodes[year]]
check[year_, round_, dept_] :=
 If[ListQ[Candidat[year, round, Region[dept]]],
  Map[check[year, round, #, dept] &, Join[{"Inscrits", "Abstentions", "Blancs ou nuls", "Exprimés", "Votants"},
    Candidat[year, round, Region[dept]]]], {True}]
check[year_, round_, res_, dept_] := DepartmentResult[year, round, res, dept] ==
  Total[Map[CommuneResult[year, round, res, #] &, Commune[year, dept]]]
```

```
check["04", "R1"];
Position[%, Except[True], {2}, Heads → False]
```

```
{}
```

```
check["04", "R2"];
Position[%, Except[True], {2}, Heads → False]
check["10", "R1"];
Position[%, Except[True], {2}, Heads → False]
check["10", "R2"];
Position[%, Except[True], {2}, Heads → False]
```

```
{}
```

```
{}
```

```
{}
```

```
{}
```

Concerning Seats

At Region level

```
CheckSeats[year_, round_, reg_] := If[ListQ[Candidat[year, round, reg]],
  Seats[year, round, reg] == Total[Map[Seats[year, round, #, reg] &, Candidat[year, round, reg]]], True]
```

```
Outer[CheckSeats, {"04", "10"}, {"R1", "R2"}, regionCodes[]];
Position[%, Except[True], {3}, Heads → False]
```

```
{}
```

Between France and Regions

```
Clear[check]
check[year_, round_] := Map[check[year, round, #] &, familyCodes[year, round]]
check[year_, "R1", res_] :=
 Seats[year, "R1", res, "FRANCE"] == Total[Map[Seats[year, "R1", res, #] &, regionCodes[year]]]
check[year_, "R2", res_] := Seats[year, "R2", res, "FRANCE"] ==
  Total[Map[Seats[year, "R2", res, #] &, Select[regionCodes[year], ListQ[Candidat[year, "R2", #]] &]]]
```

```
Outer[check[#1, #2] &, {"04", "10"}, {"R1", "R2"}];
Position[%, Except[True], {3}, Heads → False]
```

```
{}
```

Seats compare to Results

```
Map[Seats["10", "R2", #, "093"] &, Candidat["10", "R2", "093"]]
```

```
{21, 30, 72}
```

```
CompareSeats[year_, round_, region_] :=
 Map[Seats[year, round, #, region] &, Candidat[year, round, region]] - ComputedSeats[year, round, region]
```

```
Map[CompareSeats["04", "R1", #] &, regionCodes["04"]]
Map[CompareSeats["04", "R2", #] &, regionCodes["04"]]
Map[CompareSeats["10", "R1", #] &, regionCodes["10"]]
Map[CompareSeats["10", "R2", #] &, Rest[regionCodes["10"]]]
```

```
{{0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0},
 {0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0},
 {0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0},
 {0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0},
 {0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0},
 {0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}}

{{0, 0}, {0, 0, 0}, {0, 0, 0}, {0, 0, 0}, {0, 0, 0}, {0, 0, 0}, {0, 0, 0}, {0, 0, 0},
 {0, 0, 0}, {0, 0, 0}, {0, 0, 0}, {0, 0, 0}, {0, 0, 0}, {0, 0, 0}, {0, 0, 0}, {0, 0}, {0, 0}, {0, 0, 0},
 {0, 0, 0}, {0, 0, 0}, {0, 0}, {0, 0, 0}, {0, 0}, {0, 0, 0}, {0, 0, 0}, {0, 0, 0, 0, 0, 0, 0}}

{{0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0},
 {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0},
 {0, 0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0, 0},
 {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0},
 {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0},
 {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0, 0, 0}}

{{0, 0, 0}, {0, 0}, {0, 0, 0}, {0, 0}, {0, 0, 0}, {0, 0, 0}, {0, 0, 0},
 {0, 0, 0}, {0, 0}, {0, 0, 0}, {0, 0, 0}, {0, 0, 0}, {0, 0, 0}, {0, 0, 0}, {0, 0}, {0, 0, 0},
 {0, 0}, {0, 0, 0}, {0, 0}, {0, 0, 0}, {0, 0, 0}, {0, 0, 0}, {0, 0}, {0, 0, 0}, {0, 0, 0}, {0, 0, 0, 0}}
```

Modelling

Principle



ComputeSeat[



]

==



Linear Model

$$\text{ModelMatrix}\left[\left\{\ ,\ ,\ \right\}\right].\begin{pmatrix}\text{LEXG}\\ \text{LCOP}\\ \text{LSOC}\\ \text{LVEC}\\ \text{LDVG}\\ \text{LUG}\\ \text{LAUT}\\ \text{LREG}\\ \text{LCMD}\\ \text{LMAJ}\\ \text{LDVD}\\ \text{LFN}\\ \text{LEXD}\end{pmatrix} == \begin{pmatrix}\text{LUG}\\ \text{LMAJ}\\ \text{LFN}\end{pmatrix}$$

Perform a Political Analysis (or Test a Political Hypothesis)

$$\text{VoteTransfert} = \begin{pmatrix} 0.85 & 0.73 & 0.96 & 0.67 & 0.87 & 0.99 & 0.3 & 0.24 & 0.3 \\ 0 & 0.2 & 0 & 0.23 & 0.6 & 0 & 0.3 & 0.42 & 0.4 \\ 0.05 & 0 & 0 & 0 & 0.2 & 0 & 0.3 & 0.08 & 0.0 \end{pmatrix}$$

Fit the 2004 Data

```
FitData[{reg_}] := Map[FitData, Department[reg]]
FitData[dept_] := Map[FitData[dept, #] &, Map[Family["04", "R2", #] &, Candidat["04", "R2", Region[dept]]]]
FitData[dept_, fam_] := LeastSquares[Map[Function[com,
    N[Map[CommResult["04", "R1", #, com] &, Union[Map[Family["04", "R1", #] &, Candidat["04", "R1", Region[dept]]]]]] /
      CommuneResult["04", "R1", "Exprimés", com]], Commune["04", dept]], Map[
   Function[com, N[CommResult["04", "R2", fam, com] / CommuneResult["04", "R2", "Exprimés", com]]], Commune["04", dept]]]

Map[MatrixForm, FitData[{"093"}]]
```

$$\left\{\begin{pmatrix} 0.395252 & 0.667973 & -0.202941 & 0.981075 & 1.12095 & -0.0573195 & 0.194437 & 0.00196739 & 0.201617 & 0.243677 \\ 0.342051 & 0.31137 & 0.953179 & 0.054108 & -1.64409 & 0.839092 & 0.00302374 & 1.01909 & 0.656959 & 0.599738 \\ 0.262697 & 0.0206569 & 0.249762 & -0.0351826 & 1.52314 & 0.218228 & 0.80254 & -0.0210527 & 0.141425 & 0.156585 \end{pmatrix},\right.$$

$$\begin{pmatrix} 0.266177 & 0.712837 & 0.486375 & 1.05017 & 0.637951 & 0.301122 & 0.157205 & -0.0465647 & 0.589208 & -0.229219 \\ 0.523275 & 0.128195 & 0.476306 & -0.0199683 & -0.44663 & 0.498997 & 0.0394035 & 1.04083 & 0.427868 & 1.16381 \\ 0.210548 & 0.158967 & 0.0373186 & -0.0302014 & 0.808679 & 0.199881 & 0.803391 & 0.00573036 & -0.0170757 & 0.0654062 \end{pmatrix},$$

$$\begin{pmatrix} 0.404238 & 0.4682 & 1.05937 & 0.978852 & 0.0614722 & 0.222741 & 0.22263 & 0.0523181 & 0.024588 & -0.737602 \\ 0.36573 & 0.349032 & 0.774503 & 0.079083 & 0.249235 & 0.557481 & -0.00495709 & 0.97158 & 0.149678 & 1.31383 \\ 0.230032 & 0.182768 & -0.833874 & -0.057935 & 0.689293 & 0.219777 & 0.782327 & -0.023898 & 0.825734 & 0.423771 \end{pmatrix},$$

$$\begin{pmatrix} 0.409629 & 0.159711 & 0.474666 & 1.05314 & -1.27509 & 0.747786 & 0.172956 & 0.00959756 & -0.374997 & -0.627996 \\ 0.546869 & 0.367182 & 0.478622 & 0.0141782 & 2.04274 & 0.180676 & 0.00901471 & 1.00197 & 0.390182 & 1.44724 \\ 0.0435017 & 0.473107 & 0.0467121 & -0.0673193 & 0.232349 & 0.0715381 & 0.818029 & -0.0115711 & 0.984814 & 0.180751 \end{pmatrix},$$

$$\begin{pmatrix} 0.406398 & 1.11549 & 0.520541 & 1.10495 & -1.46009 & -0.439172 & 0.120666 & -0.00981914 & 0.0494071 & -0.124288 \\ 0.322398 & 0.20419 & 0.857663 & -0.0534237 & 1.8207 & 0.634145 & 0.00732516 & 1.04953 & 0.619337 & 1.03665 \\ 0.271203 & -0.319682 & -0.378204 & -0.0515249 & 0.639396 & 0.805027 & 0.872009 & -0.0397065 & 0.331255 & 0.0876412 \end{pmatrix},$$

$$\left.\begin{pmatrix} 0.234434 & 0.662638 & 0.0700846 & 1.07947 & -0.78756 & -0.285408 & 0.0914092 & -0.00779748 & 0.473535 & 0.0779128 \\ 0.367498 & 0.44318 & 1.24847 & -0.0296024 & 1.46524 & 0.703311 & -0.00876936 & 1.04853 & 0.0391685 & 0.880742 \\ 0.398068 & -0.105817 & -0.318558 & -0.0498678 & 0.32232 & 0.582097 & 0.91736 & -0.0407313 & 0.487296 & 0.0413457 \end{pmatrix}\right\}$$

Conclusion : Results

```
SlideView[
 Map[GraphicsRow[{PieChart[{Map[Last, reals[[#]]], Map[Last, expecteds[[#]]]}, ChartLegends → Map[First, reals[[#]]]],
      France[{regionCodes[][[# + 1]]}]}, ImageSize → 800] &,
  Range[Length[regionCodes[]] - 1]], AppearanceElements → { "SliderControl", "PlayPauseButton"}]
```



Code

```
Quit[]
```